

## Example 4b: General Linear Models with Multiple Fixed Effects of Multiple Predictors (complete syntax, data, and output available for SAS, STATA, and R electronically)

The models for this example come from Hoffman (2015) chapter 2. We will be examining the extent to which cognition (as measured by an information test outcome) can be predicted from age (centered at 85 years), grip strength (centered at 9 pounds), sex (with men as the reference group) and subsequent dementia diagnosis (none = 1, future = 2, and current = 3, with the none group=1 as the reference) in a sample of 550 older adults. We will first examine the bivariate relationship of each predictor with the cognition outcome (via Pearson correlations if possible or with general linear models as needed), followed by their unique effects in a simultaneous model.

### SAS Syntax for Importing and Preparing Data for Analysis:

```
* Defining global variable for file location to be replaced in code below;
* \\Client\ precedes path in Virtual Desktop outside H drive;
  %LET filesave= C:\Dropbox\22SP_PSQF6243\PSQF6243_Example4b;
* Location for SAS files for these models (uses macro variable filesave);
  LIBNAME filesave "&filesave.";

* Import chapter 2 example data into work library as Example4b;
DATA work.Example4b; SET filesave.SAS_Chapter2;
* Center quantitative predictors near their means;
  age85 = age - 85;
  grip9 = grip - 9;
* Create 2 indicator-dummy-coded binary predictors for 3 dementia groups;
  demNF=.; demNC=.; * Create two new empty variables;
  IF demgroup=1 THEN DO; demNF=0; demNC=0; END; * Replace each for none group;
  IF demgroup=2 THEN DO; demNF=1; demNC=0; END; * Replace each for future group;
  IF demgroup=3 THEN DO; demNF=0; demNC=1; END; * Replace each for current group;
* Label new variables - note semi-colon is only at the end of ALL labels;
LABEL
  age85=      "age85: Age in Years (0=85)"
  grip9=      "grip9: Grip Strength in Pounds (0=9)"
  sexMW=      "sexMW: Sex (0=M, 1=W)"
  demNF=      "demNF: Dementia Contrast for None=0 vs Future=1"
  demNC=      "demNC: Dementia Contrast for None=0 vs Current=1"
  cognition=  "cognition: Cognition Outcome"
  demgroup=   "demgroup: Dementia Group 1N 2F 3C";
* Select cases complete on all variables to be used;
  IF NMISS(cognition, age, grip, sexmw, demgroup)>0 THEN DELETE;
RUN;
```

### STATA Syntax for Importing and Preparing Data for Analysis:

```
// Defining global variable for file location to be replaced in code below
// \\Client\ precedes path in Virtual Desktop outside H drive
  cd "C:\Dropbox\22SP_PSQF6243\PSQF6243_Example4b"

// Import chapter 2 data in STATA format
  use "STATA_Chapter2.dta", clear // Has converted all variables to lower-case

// Center quantitative predictors near their means
  gen age85 = age - 85
  gen grip9 = grip - 9
// Create 2 indicator-dummy-coded binary predictors for 3 dementia groups
  gen demnf=. // Create 2 new empty variables
  gen demnc=.
// Replace for demgroup = none
  replace demnf=0 if demgroup==1
  replace demnc=0 if demgroup==1
// Replace for demgroup = future
  replace demnf=1 if demgroup==2
  replace demnc=0 if demgroup==2
```

```
// Replace for demgroup = current
replace demnf=0 if demgroup==3
replace demnc=1 if demgroup==3
// Label all variables
label variable age85      "age85: Age in Years (0=85)"
label variable grip9      "grip9: Grip Strength in Pounds (0=9)"
label variable sexmw      "sexmw: Sex (0=Men, 1=Women)"
label variable demnf      "demnf: Dementia Contrast for None=0 vs Future=1"
label variable demnc      "demnc: Dementia Contrast for None=0 vs Current=1"
label variable cognition  "cognition: Cognition Outcome"
label variable demgroup   "demgroup: Dementia Group 1N 2F 3C"
// Select cases complete on variables to be used
egen nmiss=rowmiss(cognition age grip sexmw demgroup)
drop if nmiss>0
```

## R Syntax for Importing and Preparing Data for Analysis:

```
# Set working directory (to import and export files to)
# Paste in the folder address where "SAS_Chapter2.sas7bdat" is saved in quotes
setwd("C:/Dropbox/22SP_PSQF6243/PSQF6243_Example4b")

# Import chapter 2 SAS data using haven package
Example4b = read_sas(data_file="SAS_Chapter2.sas7bdat")
# Convert to data frame to use for analysis
Example4b = as.data.frame(Example4b)

# Center quantitative predictors near their means
Example4b$age85=Example4b$age-85
Example4b$grip9=Example4b$grip-9

# Create 2 indicator-dummy-coded binary predictors for 3 dementia groups
Example4b$demNF=NA; Example4b$demNC=NA # Create 2 new empty variables
Example4b$demNF[which(Example4b$demgroup==1)]=0 # Replace each for none group
Example4b$demNC[which(Example4b$demgroup==1)]=0
Example4b$demNF[which(Example4b$demgroup==2)]=1 # Replace each for future group
Example4b$demNC[which(Example4b$demgroup==2)]=0
Example4b$demNF[which(Example4b$demgroup==3)]=0 # Replace each for current group
Example4b$demNC[which(Example4b$demgroup==3)]=1
# demNF: None=0 vs Future=1
# demNC: None=0 vs Current=1

# Select cases complete on all variables to be used
Example4b = Example4b[complete.cases(Example4b[,
  c("cognition", "age", "grip", "sexMW", "demgroup")]),]
```

## Syntax and SAS Output for Creating Descriptive Statistics:

```
TITLE "SAS Descriptive Statistics for Quantitative Variables";
PROC MEANS NONOBS NDEC=3 MEAN STDDEV VAR MIN MAX DATA=work.Example4b;
  VAR cognition age grip sexMW;
RUN;

display "STATA Descriptive Statistics for Quantitative Variables"
summarize cognition age grip sexmw, detail

print("R Descriptive Statistics for Quantitative Variables")
describe(x=Example4b[, c("cognition", "age", "grip", "sexMW")])
```

Variable	Label	Mean	Std Dev	Variance	Minimum	Maximum
cognition	cognition: Cognition Outcome	24.822	10.989	120.759	0.000	44.000
age	age: Age in Years	84.927	3.430	11.765	80.016	96.967
grip	grip: Grip Strength in Pounds	9.113	2.983	8.898	0.000	19.000
sexMW	sexMW: Sex (0=M, 1=W)	0.587	0.493	0.243	0.000	1.000

```
TITLE "SAS Descriptive Statistics for Categorical Variables";
PROC FREQ DATA=work.Example4b;
    TABLE sexMW demgroup;
RUN; TITLE;

display "STATA Descriptive Statistics for Categorical Variables"
tabulate sexmw
tabulate demgroup

print("R Descriptive Statistics for Categorical Variables")
prop.table(table(x=Example4b$sexMW,useNA="ifany"))
prop.table(table(x=Example4b$demgroup,useNA="ifany"))
```

sexMW: Sex (0=M, 1=W)				
sexMW	Frequency	Percent	Cumulative Frequency	Cumulative Percent
0	227	41.27	227	41.27
1	323	58.73	550	100.00

demgroup: Dementia Group 1N 2F 3C				
demgroup	Frequency	Percent	Cumulative Frequency	Cumulative Percent
1	399	72.55	399	72.55
2	109	19.82	508	92.36
3	42	7.64	550	100.00

**Syntax and Partial SAS Output for Creating Bivariate Pearson Correlations:**

```
TITLE "SAS Bivariate Correlations (OUT saves corrs to dataset)";
PROC CORR NOSIMPLE DATA=work.Example4b OUT=work.Corrs;
    VAR cognition age grip sexMW;
RUN; TITLE;

display "STATA Bivariate Correlations"
pwcorr cognition age grip sexmw, sig

print("R Pearson Correlation Matrix with P-values using rcorr from Hmisc package")
print("Must convert data frame to matrix to use rcorr")
rcorr(x=as.matrix(Example4b[,c("cognition","age","grip","sexMW")]), type="pearson")
```

Pearson Correlation Coefficients, N = 550 Prob >  r  under H0: Rho=0				
	cognition	age	grip	sexMW
<b>cognition</b> cognition: Cognition Outcome	1.00000	-0.17045 <.0001	0.24183 <.0001	-0.23628 <.0001
<b>age</b> age: Age in Years	-0.17045 <.0001	1.00000	-0.18414 <.0001	0.04560 0.2858
<b>grip</b> grip: Grip Strength in Pounds	0.24183 <.0001	-0.18414 <.0001	1.00000	-0.40324 <.0001
<b>sexMW</b> sexMW: Sex (0=M, 1=W)	-0.23628 <.0001	0.04560 0.2858	-0.40324 <.0001	1.00000

Note that the binary dummy codes for dementia group are not included in this correlation matrix. This is because each of their meanings would differ if the predictor were included in separate correlations than when both predictors are together in the same model. So we will need a GLM to examine the bivariate relationship between dementia group and cognition...

What can we conclude about the linear relationships between each pair of variables?

Cognition and age:

Cognition and grip:

Cognition and sex:

Age and grip:

Age and sex:

Grip and sex:

**Syntax and Partial SAS Output for Empty General Linear Model Predicting Cognition:**

$$Cognition_i = \beta_0 + e_i$$

```
TITLE "SAS Empty Model Predicting Cognition";
PROC GLM DATA=work.Example4b NAMELEN=100;
    MODEL cognition = / SOLUTION ALPHA=.05 CLPARM SS3;
RUN; QUIT; TITLE;

display "STATA Empty Model Predicting Cognition"
regress cognition, level(95)

print("R Empty Model Predicting Cognition")
ModelEmpty = lm(data=Example4b, formula=cognition~1)
anova(ModelEmpty) # anova to print residual variance
summary(ModelEmpty) # summary to print fixed effects solution
confint(ModelEmpty, level=.95) # confint for level% CI for fixed effects

# Save sums of squares from empty model for later calculations
SStotal=anova(ModelEmpty)`Sum Sq`
```

**SAS Empty Model Predicting Cognition**

Sums of Squares for “error” is SS total used in compute semi-partial R<sup>2</sup>

Source	DF	Sum of Squares	Mean Square	F Value	Pr > F
Model	1	338867.4618	338867.4618	2806.15	<.0001
Error	549	<b>66296.5382</b>	<b>120.7587</b>		
Uncorrected Total	550	405164.0000			

R-Square	Coeff Var	Root MSE	cognition Mean
<b>0.000000</b>	44.27165	10.98903	24.82182

**Mean Square Error** (Mean Square **Residual** in STATA) gives the residual variance = 120 here. In the empty model this is ALL the outcome variance.

Parameter	Estimate	Standard Error	t Value	Pr >  t	95% Confidence Limits
Intercept	24.82181818	0.46857370	52.97	<.0001	23.90140147 25.74223490 <b>Beta0</b>

Our first predictor is age—we know from the Pearson correlation matrix that there is a significant linear relationship between age and cognition, but is that linear relationship sufficient? We will estimate a GLM to check for a possible curvilinear relationship instead by including both linear and quadratic terms (age and age<sup>2</sup>, each centered so that 0 = 85 years).

**Syntax and Partial SAS Output for Linear + Quadratic Age (0=85 years) Predicting Cognition:**

$$Cognition_i = \beta_0 + \beta_1(Age_i - 18) + \beta_2(Age_i - 18)^2 + e_i$$

```
TITLE "SAS Linear + Quadratic Age (0=85) Predicting Cognition";
PROC GLM DATA=work.Example4b NAMELEN=100;
    MODEL cognition = age85 age85*age85 / SOLUTION ALPHA=.05 CLPARM SS3 EFFECTSIZE;
RUN; QUIT; TITLE;
```

```
display "STATA Linear + Quadratic Age (0=85) Predicting Cognition"
regress cognition c.age85 c.age85#c.age85, level(95)
```

```
print("R Linear + Quadratic Age (0=85) Predicting Cognition")
ModelQAge = lm(data=Example4b, formula=cognition~1+age85+I(age85^2))
anova(ModelQAge) # anova to print residual variance
summary(ModelQAge) # summary to print fixed effects solution
confint(ModelQAge, level=.95) # confint for level% CI for fixed effects
```

#### SAS Linear + Quadratic Age (0=85) Predicting Cognition

Source	DF	Sum of Squares	Mean Square	F Value	Pr > F
Model	2	1961.68396	980.84198	8.34	0.0003
Error	547	64334.85422	117.61399		
Corrected Total	549	66296.53818			

R-Square	Coeff Var	Root MSE	cognition Mean
0.029590	43.69139	10.84500	24.82182

Parameter	Estimate	Standard Error	t Value	Pr >  t	95% Confidence Limits		
Intercept	24.57135998	0.60058398	40.91	<.0001	23.39162668	25.75109327	Beta0
age85	-0.60946483	0.17752327	-3.43	0.0006	-0.95817562	-0.26075403	Beta1
age85*age85	0.01751944	0.03188742	0.55	0.5829	-0.04511735	0.08015623	Beta2

Interpret  $\beta_0$  = Intercept:

Interpret  $\beta_1$  = slope of age85:

Interpret  $\beta_2$  = slope of age85<sup>2</sup>:

Do we know if the  $R^2$  changed significantly relative to linear age only ( $r^2 = -0.17045^2 = .0291$ )?

Our second predictor is grip strength—we know from the Pearson correlation matrix that there is a significant linear relationship between grip strength and cognition, but is that linear relationship sufficient? We will estimate a GLM to check for a possible curvilinear relationship instead by including both linear and quadratic terms (grip and grip<sup>2</sup>, each centered so that 0 = 9 pounds/sq inch).

#### Syntax and Partial SAS Output for Linear + Quadratic Grip (0=9 pounds) Predicting Cognition:

$$\text{Cognition}_i = \beta_0 + \beta_1(\text{Grip}_i - 9) + \beta_2(\text{Grip}_i - 9)^2 + e_i$$

```
TITLE "SAS Linear + Quadratic Grip (0=9) Predicting Cognition";
PROC GLM DATA=work.Example4b NAMELEN=100;
    MODEL cognition = grip9 grip9*grip9 / SOLUTION ALPHA=.05 CLPARM SS3 EFFECTSIZE;
RUN; QUIT; TITLE;
```

```
display "STATA Linear + Quadratic Grip (0=9) Predicting Cognition"
regress cognition c.grip9 c.grip9#c.grip9, level(95)
```

```
print("R Linear + Quadratic Grip (0=9) Predicting Cognition")
ModelQGrip = lm(data=Example4b, formula=cognition~1+grip9+I(grip9^2))
anova(ModelQGrip) # anova to print residual variance
summary(ModelQGrip) # summary to print fixed effects solution
confint(ModelQGrip, level=.95) # confint for level% CI for fixed effects
```

## SAS Linear + Quadratic Grip (0=9) Predicting Cognition

Source	DF	Sum of Squares	Mean Square	F Value	Pr > F
Model	2	<b>3897.77368</b>	1948.88684	<b>17.08</b>	<.0001
Error	547	62398.76450	<b>114.07452</b>		
Corrected Total	549	<b>66296.53818</b>			

R-Square	Coeff Var	Root MSE	cognition Mean
<b>0.058793</b>	43.02895	10.68057	24.82182

Parameter	Estimate	Standard Error	t Value	Pr >  t	95% Confidence Limits		
Intercept	24.57890721	0.56607286	43.42	<.0001	23.46696445	25.69084997	<b>Beta0</b>
grip9	0.88761523	0.15300914	5.80	<.0001	0.58705779	1.18817266	<b>Beta1</b>
grip9*grip9	0.01606069	0.03784681	0.42	0.6715	-0.05828220	0.09040358	<b>Beta2</b>

Interpret  $\beta_0$  = Intercept:

Interpret  $\beta_1$  = slope of grip9:

Interpret  $\beta_2$  = slope of grip9<sup>2</sup>:

Do we know if the  $R^2$  changed significantly relative to linear grip only ( $r^2 = 0.24183^2 = .0585$ )?

## Syntax and Partial SAS Output with Three-Category Dementia Group Predicting Cognition:

$$\text{Cognition} = \beta_0 + \beta_1(\text{DemNF}_i) + \beta_2(\text{DemNC}_i) + e_i$$

None Mean:  $\hat{y}_N = \beta_0 + \beta_1(0) + \beta_2(0) = \beta_0 \leftarrow$  fixed effect #1

Future Mean:  $\hat{y}_F = \beta_0 + \beta_1(1) + \beta_2(0) = \beta_0 + \beta_1 \leftarrow$  linear combination

Current Mean:  $\hat{y}_C = \beta_0 + \beta_1(0) + \beta_2(1) = \beta_0 + \beta_2 \leftarrow$  linear combination

Diff of None vs. Future:  $(\beta_0 + \beta_1) - (\beta_0) = \beta_1 \leftarrow$  fixed effect #2

Diff of None vs. Current:  $(\beta_0 + \beta_2) - (\beta_0) = \beta_2 \leftarrow$  fixed effect #3

Diff of Future vs. Current:  $(\beta_0 + \beta_2) - (\beta_0 + \beta_1) = \beta_2 - \beta_1 \leftarrow$  linear combination

```
TITLE "SAS 3-Category Dementia Group (Ref=None) Predicting Cognition";
PROC GLM DATA=work.Example4b NAMELEN=100;
  MODEL cognition = demNF demNC / ALPHA=.05 CLPARM SOLUTION SS3 EFFECTSIZE;
  ESTIMATE "Cognition Mean: None"      intercept 1 demNF 0 demNC 0; * B0;
  ESTIMATE "Cognition Mean: Future"    intercept 1 demNF 1 demNC 0; * B0+B1;
  ESTIMATE "Cognition Mean: Current"   intercept 1 demNF 0 demNC 1; * B0+B2;
  ESTIMATE "Mean Diff: None vs. Future" demNF 1 demNC 0; * B1;
  ESTIMATE "Mean Diff: None vs. Current" demNF 0 demNC 1; * B2;
  ESTIMATE "Mean Diff: Future vs. Current" demNF -1 demNC 1; * B2-B1;
  ODS OUTPUT Estimates=work.DemEstimates; * Save for computing effect sizes;
RUN; QUIT; TITLE;
```

```
* SAS code to compute effect sizes from stored ESTIMATE results;
DATA work.DemEstimates; SET work.DemEstimates;
* Cohen d is partial standardized mean difference;
  PartialD=(2*tValue)/SQRT(547); * SQRT(number) = DF denominator;
* PartialR is partial correlation;
  PartialR = tvalue/(SQRT(tvalue**2 +547)); * +number = DF denominator;
RUN;
```

```
* Print estimates table with effect sizes added;
TITLE "Effect Sizes for 3-Category DemGroup";
PROC PRINT NOOBS DATA=work.DemEstimates;
  VAR Parameter Estimate StdErr probt PartialD PartialR;
RUN; TITLE;
```

```

display "STATA 3-Category Dementia Group (Ref=None) Predicting Cognition"
regress cognition c.demnf c.demnc, level(95)
lincom _cons*1 + c.demnf*0 + c.demnc*0 // Cognition Mean for None = B0
lincom _cons*1 + c.demnf*1 + c.demnc*0 // Cognition Mean for Future = B0+B1
lincom _cons*1 + c.demnf*0 + c.demnc*1 // Cognition Mean for Current = B0+B2
lincom c.demnf*1 + c.demnc*0 // Mean Diff: None vs. Future = B1
    display "PartialD= " (2*(r(estimate)/r(se)))/sqrt(r(df))
    display "PartialR= " (r(estimate)/r(se))/sqrt((r(estimate)/r(se))^2+r(df))
lincom c.demnf*0 + c.demnc*1 // Mean Diff: None vs. Current = B2
    display "PartialD= " (2*(r(estimate)/r(se)))/sqrt(r(df))
    display "PartialR= " (r(estimate)/r(se))/sqrt((r(estimate)/r(se))^2+r(df))
lincom c.demnf*-1 + c.demnc*1 // Mean Diff: Future vs. Current = B2-B1
    display "PartialD= " (2*(r(estimate)/r(se)))/sqrt(r(df))
    display "PartialR= " (r(estimate)/r(se))/sqrt((r(estimate)/r(se))^2+r(df))

print("R 3-Category Dementia Group (Ref=None) Predicting Cognition")
ModelDem = lm(data=Example4b, formula=cognition~1+demNF+demNC)
anova(ModelDem) # anova to print residual variance
summary(ModelDem) # summary to print fixed effects solution
confint(ModelDem, level=.95) # confint for level% CI for fixed effects

print("R Ask for predicted cognition per group and group differences")
print("In number lists below, values are multiplier for each fixed effect IN ORDER")
PredDem = glht(model=ModelDem, linfct=rbind(
  "Cognition Mean: None" = c(1, 0, 0),
  "Cognition Mean: Future" = c(1, 1, 0),
  "Cognition Mean: Current" = c(1, 0, 1),
  "Mean Diff: None vs Future" = c(0, 1, 0),
  "Mean Diff: None vs Current" = c(0, 0, 1),
  "Mean Diff: Future vs Current" = c(0,-1, 1)))
print("Save glht linear combination results with unadjusted p-values and 95% CIs")
SavePredDem = summary(PredDem, test=adjusted("none"))
print(SavePredDem); confint(PredDem, level=.95, calpha=univariate_calpha())

# R code to compute effect sizes from stored GLHT results
PredDemPartialD=(2*SavePredDem$test$tstat)/sqrt(ModelDem$df.residual)
PredDemPartialR=SavePredDem$test$tstat/
  sqrt(SavePredDem$test$tstat^2+ModelDem$df.residual)
# Concatenate effect sizes to GLHT table for mean differences
data.frame(Estimate=SavePredDem$test$coefficients, SE=SavePredDem$test$sigma,
  pvalue=SavePredDem$test$pvalues,
  PartialD=PredDemPartialD, PartialR=PredDemPartialR)

```

SAS 3-Category Dementia Group (Ref=None) Predicting Cognition

Source	DF	Sum of Squares	Mean Square	F Value	Pr > F
Model	2	<b>11685.51093</b>	5842.75547	<b>58.52</b>	<.0001
Error	547	54611.02725	<b>99.83734</b>		
Corrected Total	549	<b>66296.53818</b>			

R-Square	Coeff Var	Root MSE	cognition Mean
<b>0.176261</b>	40.25436	9.991864	24.82182

Table of Model-Estimated Fixed Effects (normally is last)

Parameter	Estimate	Standard Error	t Value	Pr >  t	95% Confidence Limits		
Intercept	27.19799499	0.50021886	54.37	<.0001	26.21540992	28.18058005	<b>Beta0</b>
demNF	-5.67505921	1.07988789	-5.26	<.0001	-7.79629411	-3.55382430	<b>Beta1</b>
demNC	-16.38847118	1.62089436	-10.11	<.0001	-19.57241068	-13.20453167	<b>Beta2</b>

Interpret  $\beta_0$  = Intercept:

Interpret  $\beta_1$  = slope of None vs. Future:

Interpret  $\beta_2$  = slope of None vs. Current:

Table of Extra Requested Linear Combinations of Model-Estimated Fixed Effects

Parameter	Estimate	Standard Error	t Value	Pr >  t	95% Confidence Limits	
Cognition Mean: None	27.1979950	0.50021886	54.37	<.0001	26.2154099	28.1805801
Cognition Mean: Future	21.5229358	0.95704699	22.49	<.0001	19.6429985	23.4028730
Cognition Mean: Current	10.8095238	1.54177807	7.01	<.0001	7.7809932	13.8380544
Mean Diff: None vs. Future	-5.6750592	1.07988789	-5.26	<.0001	-7.7962941	-3.5538243
Mean Diff: None vs. Current	-16.3884712	1.62089436	-10.11	<.0001	-19.5724107	-13.2045317
Mean Diff: Future vs. Current	-10.7134120	1.81466762	-5.90	<.0001	-14.2779823	-7.1488417

Table with Effect Sizes Added:

Parameter	Estimate	StdErr	Probt	PartialD	PartialR
Cognition Mean: None	27.1979950	0.50021886	<.0001	4.64957	0.91862
Cognition Mean: Future	21.5229358	0.95704699	<.0001	1.92311	0.69312
Cognition Mean: Current	10.8095238	1.54177807	<.0001	0.59954	0.28715
Mean Diff: None vs. Future	-5.6750592	1.07988789	<.0001	-0.44939	-0.21923
Mean Diff: None vs. Current	-16.3884712	1.62089436	<.0001	-0.86461	-0.39681
Mean Diff: Future vs. Current	-10.7134120	1.81466762	<.0001	-0.50486	-0.24475

### Syntax and Partial Output by Program with All 4 Predictors of Cognition

(with requested F-tests to demonstrate how to test the unique joint contributions of 2+ slopes simultaneously, as in "hierarchical" or "stepwise" regression):

$$\text{Cognition}_i = \beta_0 + \beta_1(\text{Age}_i - 85) + \beta_2(\text{Grip}_i - 9) + \beta_3(\text{SexMW}_i) + \beta_4(\text{DemNF}_i) + \beta_5(\text{DemNC}_i) + e_i$$

```
TITLE "SAS Combined Model Predicting Cognition";
PROC GLM DATA=work.Example4b NAMELEN=100;
MODEL cognition = age85 grip9 sexMW demNF demNC / ALPHA=.05 CLPARM SOLUTION SS3 EFFECTSIZE;
CONTRAST "F-Test of DFnum=5 for Model" age85 1, grip9 1, sexMW 1, demNF 1, demNC 1;
CONTRAST "F-Test DFnum=2 for Demgroup" demNF 1, demNC 1; * For overall group;
CONTRAST "F-Test DFnum=2 for Age and Sex" age85 1, sexMW 1; * For demographic vars;
ESTIMATE "Mean Diff: Future vs. Current" demNF -1 demNC 1; * B5-B4;
ODS OUTPUT ParameterEstimates=work.ModelEstimates Estimates=work.ReqEstimates;
RUN; QUIT; TITLE; * Save for computing effect sizes;

* Compute Cohen d effect size and r effect sizes;
DATA work.FinalEstimates; LENGTH Parameter $50;
SET work.ModelEstimates ReqEstimates;
* Cohen d is partial standardized mean difference;
PartialD=(2*tValue)/SQRT(544); * SQRT(number) = DF denominator;
* PartialR is partial correlation;
PartialR = tvalue/(SQRT(tvalue**2 +544)); * +number = DF denominator;
RUN;
* Print solution and estimates combined table with effect sizes;
TITLE "Effect Sizes for Final Model";
PROC PRINT NOOBS DATA=work.FinalEstimates;
VAR Parameter Estimate StdErr probt PartialD PartialR;
RUN; TITLE;

display "STATA Combined Model Predicting Cognition"
regress cognition c.age85 c.grip9 c.sexmw c.demnf c.demnc, level(95)
test (c.age85=0) (c.grip9=0) (c.sexmw=0) (c.demnf=0) (c.demnc=0) // F-Test DFnum=5 for Model
test (c.demnf=0) (c.demnc=0) // F-Test DFnum=2 for Demgroup (for overall group)
test (c.age85=0) (c.sexmw=0) // F-Test DFnum=2 for Age and Sex (for demographic vars)
lincom c.demnf*1 + c.demnc*0 // Mean Diff: None vs. Future = B4
display "PartialD= " (2*(r(estimate)/r(se)))/sqrt(r(df))
display "PartialR= " (r(estimate)/r(se))/sqrt((r(estimate)/r(se))^2+r(df))
```



```

lincom c.demnf*0 + c.demnc*1 // Mean Diff: None vs. Current = B5
    display "PartialD= " (2*(r(estimate)/r(se)))/sqrt(r(df))
    display "PartialR= " (r(estimate)/r(se))/sqrt((r(estimate)/r(se))^2+r(df))
lincom c.demnf*-1 + c.demnc*1 // Mean Diff: Future vs. Current = B5-B4
    display "PartialD= " (2*(r(estimate)/r(se)))/sqrt(r(df))
    display "PartialR= " (r(estimate)/r(se))/sqrt((r(estimate)/r(se))^2+r(df))
// Get partial correlations for age, grip, and sex (sr2 for dem slopes are not valid)
pcorr cognition c.age85 c.grip9 c.sexmw c.demnf c.demnc

// Stata: calculate semipartial eta-squared for demgroup using SS from reduced model
display "STATA Reduced Model to Get SS for demnf and demnc (as omitted)"
regress cognition c.age85 c.grip9 c.sexmw, level(95)
// sr2 = (SSfull-SSreduced)/SStotal
display (18385.97930-6574.67775)/66296.5382 // sr2 for demgroup = .17815865

// Stata: calculate semipartial eta-squared for age and sex using SS from reduced model
display "STATA Reduced Model to Get SS for age85 and sexmw (as omitted)"
regress cognition c.grip9 c.demnf c.demnc, level(95)
// sr2 = (SSfull-SSreduced)/SStotal
display (18385.97930-15956.7691)/66296.5382 // sr2 for age and sex = .03664158

```

Note the new use of “Anova” in addition to “anova” in R... seriously...

```

print("R Combined Model Predicting Cognition")
ModelAll = lm(data=Example4b, formula=cognition~1+age85+grip9+sexMW+demNF+demNC)
anova(ModelAll) # anova to print residual variance
Anova(ModelAll, Type="3") # Anova from car package to get Type 3 SS
SumModelAll = summary(ModelAll); SumModelAll # save summary, print fixed effects solution
confint(ModelAll, level=.95) # confint for level% CI for fixed effects

print("R Ask for missing model-implied group difference")
PredModelAll = glht(model=ModelAll, linfct=rbind("Future vs Current" = c(0,0,0,0,-1,1)))
print("Save glht linear combination results with unadjusted p-values and 95% CIs")
SaveModelAll = summary(PredModelAll, test=adjusted("none"))
print(SaveModelAll); confint(PredModelAll, level=.95, calpha=univariate_calpha())

print("Replicate F-test for Model: includes all 5 slopes")
ModelAllFall = glht(model=ModelAll,
    linfct=c("age85=0", "grip9=0", "sexMW=0", "demNF=0", "demNC=0"))
summary(ModelAllFall, test=Ftest()) # ask for joint hypothesis test instead of separate

print("Ask for F-test for overall effect of DemGroup")
ModelAllFdem = glht(model=ModelAll, linfct=c("demNF=0", "demNC=0"))
summary(ModelAllFdem, test=Ftest()) # ask for joint hypothesis test instead of separate

print("R Semi-Partial R2 for DemGroup via SS for Reduced Model Omitting DemGroup")
ModelNoDem = lm(data=Example4b, formula=cognition~1+age85+grip9+sexMW)
SSDem=anova(ModelNoDem, ModelAll); SSDem$`Sum of Sq`/SStotal

print("Ask for F-test for effects of Age and Sex")
ModelAllFagesex = glht(model=ModelAll, linfct=c("age85=0", "sexMW=0"))
summary(ModelAllFagesex, test=Ftest()) # ask for joint hypothesis test instead of separate

print("R Semi-Partial R2 for Age and Sex via SS for Reduced Model Omitting Age and Sex")
ModelAllNoAgeSex = lm(data=Example4b, formula=cognition~1+grip9+demNF+demNC)
SSAgeSex=anova(ModelAllNoAgeSex, ModelAll); SSAgeSex$`Sum of Sq`/SStotal

print("R code to compute effect sizes from stored model results from GLHT")
DemPartialD=(2*SaveModelAll$test$tstat)/sqrt(ModelAll$df.residual)
DemPartialR=SaveModelAll$test$tstat/sqrt(SaveModelAll$test$tstat^2+ModelAll$df.residual)
# Concatenate effect sizes to results table for mean differences
data.frame(Estimate=SaveModelAll$test$coefficients, SE=SaveModelAll$test$sigma,
    pvalue=SaveModelAll$test$pvalues,
    PartialD=DemPartialD, PartialR=DemPartialR)

```

```
print("R code to compute effect sizes from stored model results from fixed effects")
ModelAllpD=(2*SumModelAll$coefficients[,"t value"])/sqrt(ModelAll$df.residual)
ModelAllpR= SumModelAll$coefficients[,"t value"]/
sqrt(SumModelAll$coefficients[,"t value"]^2+ModelAll$df.residual)
# Concatenate effect sizes to results table for fixed effects
data.frame(SumModelAll$coefficients, PartialD=ModelAllpD, PartialR=ModelAllpR)
```

**Partial SAS Output:**

**SAS Combined Model Predicting Cognition**

Source	DF	Sum of Squares	Mean Square	F Value	Pr > F
Model	5	<b>18385.97930</b>	3677.19586	<b>41.75</b>	<.0001
Error	544	47910.55888	<b>88.07088</b>		
Corrected Total	549	<b>66296.53818</b>			

R-Square	Coeff Var	Root MSE	cognition Mean
<b>0.277329</b>	37.80790	9.384609	24.82182

Source	DF	Type III SS	Mean Square	F Value	Pr > F
age85	1	1025.58587	1025.58587	11.65	0.0007
grip9	1	1433.33570	1433.33570	16.27	<.0001
sexMW	1	1482.49792	1482.49792	16.83	<.0001
demNF	1	2776.56843	2776.56843	31.53	<.0001
demNC	1	10315.20016	10315.20016	117.12	<.0001

Total Variation Accounted For  
Semipartial

Source	Semipartial Eta-Square	Omega-Square	Conservative 95% Confidence Limits	
age85	<b>0.0155</b>	0.0141	0.0017	0.0418
grip9	<b>0.0216</b>	0.0203	0.0041	0.0512
sexMW	<b>0.0224</b>	0.0210	0.0044	0.0523
demNF	0.0419	0.0405	0.0152	0.0791 → Do not report (not valid)
demNC	0.1556	0.1541	0.1041	0.2100 → Do not report (not valid)

$$sr^2 = SS_{slope} / SS_{total}$$

Contrast	DF	Contrast SS	Mean Square	F Value	Pr > F
F-test of DFnum=5 for Model	5	18385.97930	3677.19586	41.75	<.0001 → for model
F-Test DFnum=2 for Demgroup	2	11811.30155	5905.65077	67.06	<.0001 → for demgroup
F-Test DFnum=2 for Age and Sex	2	2429.21021	1214.60511	13.79	<.0001 → for demog vars

Total Variation Accounted For  
Semipartial

Contrast	Semipartial Eta-Square	Omega-Square	Conservative 95% Confidence Limits	
F-Test DFnum=5 for Model (demo)	<b>0.2773</b>	0.2703	0.2116	0.3299 ← 18385.98/66296.54
F-Test DFnum=2 for Demgroup	<b>0.1782</b>	0.1753	0.1227	0.2325 ← 11811.30/66296.54
F-Test DFnum=2 for Age and Sex	<b>0.0366</b>	0.0339	0.0106	0.0702 ← 2429.21/66296.54

**Table of Model-Estimated Fixed Effects (normally is last)**

Parameter	Estimate	Standard Error	t Value	Pr >  t	95% Confidence Limits		
Intercept	29.26432541	0.69850792	41.90	<.0001	27.8922232	30.63642850	<b>Beta0</b>
age85	-0.40573396	0.11889717	-3.41	0.0007	-0.63928775	-0.17218017	<b>Beta1</b>
grip9	0.60422556	0.14977568	4.03	<.0001	0.31001605	0.89843507	<b>Beta2</b>
sexMW	-3.65737421	0.89143262	-4.10	<.0001	-5.40844590	-1.90630252	<b>Beta3</b>
demNF	-5.72197100	1.01907848	-5.61	<.0001	-7.72378184	-3.72016016	<b>Beta4</b>
demNC	-16.47981327	1.52275357	-10.82	<.0001	-19.47101037	-13.48861616	<b>Beta5</b>

**Table of Extra Requested Linear Combinations of Model-Estimated Fixed Effects**

Parameter	Estimate	Standard Error	t Value	Pr >  t
Mean Diff: Future vs. Current	-10.7578423	1.70795708	-6.30	<.0001

**Concatenated Table with Effect Sizes Added:**

Parameter	Estimate	StdErr	Probt	PartialD	PartialR
Intercept	29.26432541	0.69850792	<.0001	3.59251	0.87373
age85	-0.40573396	0.11889717	0.0007	-0.29262	-0.14477
grip9	0.60422556	0.14977568	<.0001	0.34593	0.17043
sexMW	-3.65737421	0.89143262	<.0001	-0.35181	-0.17325
demNF	-5.72197100	1.01907848	<.0001	-0.48147	-0.23405
demNC	-16.47981327	1.52275357	<.0001	-0.92801	-0.42090
Mean Diff: Future vs. Current	-10.75784226	1.70795708	<.0001	-0.54011	-0.26071

**Partial STATA Output:**

STATA Combined Model Predicting Cognition

Source	SS	df	MS	Number of obs	=	550
Model	18385.9793	5	3677.19586	<b>F(5, 544)</b>	=	<b>41.75</b>
Residual	47910.5589	544	<b>88.0708803</b>	Prob > F	=	0.0000
Total	66296.5382	549	120.758722	<b>R-squared</b>	=	<b>0.2773</b>
				Adj R-squared	=	0.2707
				Root MSE	=	9.3846

cognition	Coef.	Std. Err.	t	P> t	[95% Conf. Interval]
age85	-.405734	.1188972	-3.41	0.001	-.6392878 - .1721802
grip9	.6042256	.1497757	4.03	0.000	.310016 .8984351
sexmw	-3.657374	.8914326	-4.10	0.000	-5.408446 -1.906303
demnf	-5.721971	1.019078	-5.61	0.000	-7.723782 -3.72016
demnc	-16.47981	1.522754	-10.82	0.000	-19.47101 -13.48862
_cons	29.26433	.6985079	41.90	0.000	27.89222 30.63643

```
. test (c.age85=0)(c.grip9=0)(c.sexmw=0)(c.demnf=0)(c.demnc=0) // F-Test DFnum=5 for Model
( 1) age85 = 0
( 2) grip9 = 0
( 3) sexmw = 0
( 4) demnf = 0
( 5) demnc = 0
      F( 5, 544) = 41.75
      Prob > F = 0.0000
```

```
. test (c.demnf=0)(c.demnc=0) // F-Test DFnum=2 for Demgroup (for overall demgroup)
( 1) demnf = 0
( 2) demnc = 0
      F( 2, 544) = 67.06
      Prob > F = 0.0000
```

```
. test (c.age85=0)(c.sexmw=0) // F-Test DFnum=2 for Age and Sex (for demographic vars)
( 1) age85 = 0
( 2) sexmw = 0
      F( 2, 544) = 13.79
      Prob > F = 0.0000
```

```
. lincom c.demnf*1 + c.demnc*0 // Mean Diff: None vs. Future = B4
```

cognition	Coef.	Std. Err.	t	P> t	[95% Conf. Interval]
(1)	-5.721971	1.019078	-5.61	0.000	-7.723782 -3.72016

```
. display "Partial D= " (2*(r(estimate)/r(se))/sqrt(r(df)))
Partial D= -.48146926
. display "Partial R= " (r(estimate)/r(se))/sqrt((r(estimate)/r(se))^2+r(df))
Partial R= -.23404822
```

```
. lincom c.demnf*0 + c.demnc*1 // Mean Diff: None vs. Current = B5
-----+-----
cognition |      Coef.   Std. Err.      t    P>|t|     [95% Conf. Interval]
-----+-----
(1) |    -16.47981   1.522754    -10.82   0.000   -19.47101   -13.48862
-----+-----
```

```
. display "Partial D= "      (2*(r(estimate)/r(se)))/sqrt(r(df))
Partial D= -.92801118
. display "Partial R= "      (r(estimate)/r(se))/sqrt((r(estimate)/r(se))^2+r(df))
Partial R= -.42090225
```

```
. lincom c.demnf*-1 + c.demnc*1 // Mean Diff: Future vs. Current = B5-B4
-----+-----
cognition |      Coef.   Std. Err.      t    P>|t|     [95% Conf. Interval]
-----+-----
(1) |    -10.75784   1.707957     -6.30   0.000   -14.11284   -7.402844
-----+-----
```

```
. display "Partial D= "      (2*(r(estimate)/r(se)))/sqrt(r(df))
Partial D= -.54010571
. display "Partial R= "      (r(estimate)/r(se))/sqrt((r(estimate)/r(se))^2+r(df))
Partial R= -.26071341
```

Partial and semipartial correlations of cognition with

Variable	Partial Corr.	Semipartial Corr.	Partial Corr.^2	Semipartial Corr.^2	Significance Value
age85	-0.1448	-0.1244	0.0210	<b>0.0155</b>	0.0007
grip9	0.1704	0.1470	0.0290	<b>0.0216</b>	0.0001
sexmw	-0.1732	-0.1495	0.0300	<b>0.0224</b>	0.0000
demnf	-0.2340	-0.2046	0.0548	0.0419	0.0000 → do not use sr2
demnc	-0.4209	-0.3945	0.1772	0.1556	0.0000 → do not use sr2

STATA Reduced Model to Get SS for demnf and demnc (as omitted)

```
. regress cognition c.age85 c.grip9 c.sexmw, level(95)
-----+-----
Source |      SS          df           MS       Number of obs   =      550
-----+-----+-----+-----
Model |    6574.67775         3    2191.55925   F(3, 546)       =      20.04
Residual |   59721.8604       546    109.380697   Prob > F        =      0.0000
-----+-----+-----+-----
Total |   66296.5382       549    120.758722   R-squared       =      0.0992
                                           Adj R-squared  =      0.0942
                                           Root MSE      =      10.459
```

```
. // sr2 = (SSfull-SSreduced)/SStotal
. display (18385.97930-6574.67775)/66296.5382 // sr2 for demgroup = .17815865
.17815865
```

STATA Reduced Model to Get SS for age85 and sexmw (as omitted)

```
. regress cognition c.grip9 c.demnf c.demnc, level(95)
-----+-----
Source |      SS          df           MS       Number of obs   =      550
-----+-----+-----+-----
Model |   15956.7691         3    5318.92303   F(3, 546)       =      57.69
Residual |   50339.7691       546     92.1973793   Prob > F        =      0.0000
-----+-----+-----+-----
Total |   66296.5382       549    120.758722   R-squared       =      0.2407
                                           Adj R-squared  =      0.2365
                                           Root MSE      =      9.6019
```

```
. // sr2 = (SSfull-SSreduced)/SStotal
. display (18385.97930-15956.7691)/66296.5382 // sr2 for age and sex = .03664158
.03664158
```

**Partial R Output:**

```
> anova(ModelAll)
Analysis of Variance Table

Response: cognition
      Df Sum Sq Mean Sq F value    Pr(>F)
age85   1  1926.2   1926.18  21.8708 0.0000036833209
grip9   1  3039.2   3039.17  34.5082 0.000000073976
sexMW   1  1609.3   1609.32  18.2731 0.0000226023607
demNF   1  1496.1   1496.10  16.9875 0.0000434979953
demNC   1 10315.2 10315.20 117.1239 < 2.22e-16
Residuals 544 47910.6    88.07 → residual variance
```

```
> Anova(ModelAll, Type = "3")
Anova Table (Type II tests)
```

```
Response: cognition
      Sum Sq Df F value    Pr(>F)
age85   1025.6  1  11.6450  0.00069165
grip9   1433.3  1  16.2748 0.000062630407
sexMW   1482.5  1  16.8330 0.000047069291
demNF   2776.6  1  31.5265 0.000000031404
demNC  10315.2  1 117.1239 < 2.22e-16
Residuals 47910.6 544
```

```
Coefficients:
      Estimate Std. Error t value    Pr(>|t|)
(Intercept) 29.26433    0.69851  41.8955 < 2.2e-16
age85       -0.40573    0.11890  -3.4125  0.0006917
grip9        0.60423    0.14978   4.0342 0.0000626304
sexMW       -3.65737    0.89143  -4.1028 0.0000470693
demNF       -5.72197    1.01908  -5.6148 0.0000000314
demNC      -16.47981    1.52275 -10.8224 < 2.2e-16
```

```
Residual standard error: 9.3846 on 544 degrees of freedom
Multiple R-squared:  0.27733, Adjusted R-squared:  0.27069
F-statistic: 41.753 on 5 and 544 DF, p-value: < 2.22e-16
```

Simultaneous Tests for General Linear Hypotheses

```
Linear Hypotheses:
      Estimate Std. Error t value    Pr(>|t|)
Future vs Current == 0 -10.758    1.708 -6.2987 0.0000000006198
(Adjusted p values reported -- none method)
```

```
[1] "Replicate F-test for Model: includes all 5 slopes"
```

```
Linear Hypotheses:
      Estimate
age85 == 0 -0.40573
grip9 == 0  0.60423
sexMW == 0 -3.65737
demNF == 0 -5.72197
demNC == 0 -16.47981
Global Test:
      F DF1 DF2    Pr(>F)
1  41.753  5 544 2.1562e-36
```

```
[1] "Ask for F-test for overall effect of DemGroup"
```

```
Linear Hypotheses:
      Estimate
demNF == 0 -5.722
demNC == 0 -16.480
Global Test:
      F DF1 DF2    Pr(>F)
1  67.056  2 544 9.3117e-27
```

```
[1] "R Semi-Partial R2 for DemGroup via SS for Reduced Model Omitting DemGroup"
```

```
> ModelNoDem = lm(data = Example4b, formula = cognition ~ 1 + age85 + + grip9 + sexMW)
> SSDem = anova(ModelNoDem, ModelAll)
> SSDem$`Sum of Sq`/SStotal
[1] NA 0.17815865
```

This is the typical “anova” table we’ve been using to get the model residual variance. We should ignore the rest of it—for each slope, it provides “Type 1” sequential sums of squares, which means it credits the slopes entered first *with common predictor variance!* That means the *p*-values do not match those of the model results here...

Instead, we want “Type 3” sums of squares (which simplifies to Type 2 without interaction terms), as given by the “Anova” routine in the car package. These could be used to compute effect sizes for individual slopes. But Anova doesn’t give MSE (residual variance), so we still need anova.... Sigh.

```
[1] "Ask for F-test for effects of Age and Sex"
```

```
Linear Hypotheses:
```

```
      Estimate
age85 == 0 -0.40573
sexMW == 0 -3.65737
```

```
Global Test:
```

```
      F DF1 DF2      Pr(>F)
1 13.791   2 544 0.000001437
```

```
[1] "R Semi-Partial R2 for Age and Sex via SS for Reduced Model Omitting Age and Sex"
> ModelAllNoAgeSex = lm(data = Example4b, formula = cognition ~ + 1 + grip9 + demNF + demNC)
> SSAgeSex = anova(ModelAllNoAgeSex, ModelAll)
> SSAgeSex$`Sum of Sq`/SStotal
[1]      NA 0.036641585
```

```
[1] "R code to compute effect sizes from stored model results from GLHT"
```

```
      Estimate      SE      pvalue      PartialD      PartialR
Future vs Current -10.757842 1.7079571 0.00000000061976024 -0.54010571 -0.26071341
```

```
[1] "R code to compute effect sizes from stored model results from fixed effects"
```

```
      Estimate Std..Error      t.value      Pr...t...      PartialD      PartialR
(Intercept) 29.26432541 0.69850792 41.8954812 2.1121910e-172 3.59250787 0.87372718
age85       -0.40573396 0.11889717 -3.4124779 6.9165032e-04 -0.29261756 -0.14476752
grip9       0.60422556 0.14977568 4.0342034 6.2630407e-05 0.34593009 0.17043440
sexMW      -3.65737421 0.89143262 -4.1028050 4.7069291e-05 -0.35181263 -0.17324635
demNF      -5.72197100 1.01907848 -5.6148482 3.1403677e-08 -0.48146926 -0.23404822
demNC     -16.47981327 1.52275357 -10.8223771 7.4492710e-25 -0.92801118 -0.42090225
```

### **Example Results Section Using SAS Output [notes about what also to include]:**

Table 1 provides descriptive statistics and Pearson bivariate correlations among the cognition outcome and predictors of age, grip strength, and sex (0 = men, 1= women). As shown, cognition was predicted to be significantly greater in persons who were younger, who were stronger, and in men. To provide a meaningful intercept in the models that follow, we centered age such that 0 = 85 years and grip strength such that 0 = 9 pounds per square inch.

**Table 1: Descriptive Statistics and Bivariate Correlations (bold values indicate  $p < .0001$ )**

	Cognition	Age	Grip	Sex
Cognition	1.000			
Age	<b>-0.170</b>	1.000		
Grip Strength	<b>0.242</b>	<b>-0.184</b>	1.000	
Sex (0=Men)	<b>-0.236</b>	0.046	<b>-0.403</b>	1.000
Mean	24.822	84.927	9.113	0.587
SD	10.989	3.430	2.983	0.493
Min	0.000	80.016	0.000	0.000
Max	44.000	96.967	19.000	1.000

In separate linear regressions, we examined the potential for a quadratic effect of age and for a quadratic effect of grip strength. Neither quadratic slope was significant [could give  $p$ -values or effect sizes for completeness], indicating that linear slopes for age and grip strength were likely to be sufficient. In an analysis of variance, we examined the bivariate effect of dementia group (none = 72.55%, future = 19.82%, or current = 7.64%) in predicting cognition. We found significant mean differences in cognition across the three groups,  $F(2, 547) = 58.52$ ,  $MSE = 99.84$ ,  $p < .0001$ ,  $R^2 = .176$ . Results (including  $d$  partial effect sizes for mean differences in standard deviation units) were as follows. Relative to the reference group of no dementia ( $M = 27.198$ ,  $SE = 0.500$ ), cognition was significantly lower by 5.675 ( $SE = 1.0799$ ,  $d = -0.449$ ) in the future group ( $M = 21.522$ ,  $SE = 0.957$ ) and significantly lower by 16.388 ( $SE = 1.621$ ,  $d = -0.865$ ) in the current group ( $M = 10.810$ ,  $SE =$

1.542). Cognition in the current group was also significantly lower than the future group by 10.713 (SE = 1.815,  $d = -0.505$ ).

Finally, we combined the effects of all four predictors in the same model (i.e., an analysis of covariance or multiple linear regression) as shown in Table 2. Semipartial squared correlation ( $sr^2$ ) effect sizes were also obtained to describe the amount of variance captured by distinct sets of predictor slopes. We found that the unique contribution of each predictor remained significant in the same direction as their bivariate effects, such that cognition was predicted to be significantly higher in participants who were younger, who were stronger, in men, and in persons without dementia. The model accounted for a significant amount of variance in cognition,  $F(5, 544) = 41.75$ ,  $MSE = 88.07$ ,  $p < .0001$ ,  $R^2 = .277$ , and the omnibus effect of dementia group remained significant,  $F(2, 544) = 67.06$ ,  $p < .0001$ ,  $sr^2 = .178$ . In addition, the demographic variables of age and sex had a smaller but significant unique joint contribution,  $F(2, 544) = 13.79$ ,  $p < .0001$ ,  $sr^2 = .034$ , indicating that controlling for dementia status and grip strength did not remove their effects.

[Emphasize why it matters based on your research questions that the predictors had significant unique effects.]

**Table 2: Results from Combined Model Predicting Cognition**

Fixed Effect	Est	SE	$p <$	$d$	$r$	$sr^2$
Intercept	29.264	0.699				
Age (0=85 years)	-0.406	0.119	.001	-0.293	-0.145	0.016
Grip (0=9 pounds)	0.604	0.150	.001	0.346	0.170	0.022
Sex (0=Men)	-3.657	0.891	.001	-0.352	-0.173	0.022
None vs Future Dementia	-5.722	1.019	.001	-0.481	-0.234	
None vs Current Dementia	-16.480	1.523	.001	-0.928	-0.421	0.178
Future vs Current Dementia	-10.758	1.708	.001	-0.540	-0.261	

Note:  $d$  and  $r$  partial effect sizes were computed from the slope  $t$  test-statistics as follows:  $d = \frac{2t}{\sqrt{DF_{den}}}$ ;  $r = \frac{t}{\sqrt{t^2 + DF_{den}}}$ .

Semi-partial squared correlations ( $sr^2$ ) were computed using Type III sums of squares for each slope separately for age, grip strength, and sex, and for the joint combination of the two slopes for dementia group given their common reference group.